
Formule(s) variabel maken.

De functie indirect geeft als resultaat de verwijzing die door een tekenreeks wordt aangegeven. 
Verwijzingen worden onmiddellijk geëvalueerd, zodat de inhoud kan worden weergegeven. Met 
INDIRECT() kun je binnen een formule de verwijzing naar een cel wijzigen, zonder de formule zelf te 
wijzigen.

Wanneer je een verwijzing naar een cel maakt (bijvoorbeeld: =A5)wordt de verwijzing naar die cel 
altijd bijgewerkt wanneer je: de cel knipt (cel wordt verwijderd) of in het geval er rij(en) en of 
kolom(men) worden ingevoegd. Wil je altijd naar hetzelfde celadres laten verwijzen dan gebruik je de 
functie INDIRECT(), dus: =INDIRECT("A5").

Bijna zonder uitzondering gebruik je genoemde functie binnen een of meer andere functies, het 
zogenaamde nesten. Nesten komt er op neer dat je een functie gebruikt als argument van een 
andere functie. Hieronder vind je daar voorbeelden van. Zoals je zult begrijpen zijn de hier gegeven 
formules bedoeld om inzicht te krijgen in de functie. Het toepassen van de functie in een specifieke 
spreadsheet vraagt dan ook om een specifieke benadering.

In de kolommen A, B en C staan willekeurige getallen, in kolom D bevinden zich de formules. In 
kolom G vind je de variabelen. Eventueel maak je dit tabelletje na en kopieer je de gegeven formules 
naar de juiste cellen. Speel met de variabelen om meer inzicht in de functie INDIRECT() te krijgen.

De formule in:
• D2: =INDIRECT("A"&G2) Verwijst altijd naar cel A4, ook als deze cel verplaatst wordt.
• D3: =INDIRECT("A"&$G$2)  Eerste argument is de string kolom A & de variabele uit G2, 

verwijst naar: A1 (1001)
• D4: =SOM(INDIRECT($G$4&$G$2):INDIRECT($G$4&$G$3))  Telt de waarden op in kolom B 

van de rijen 2+3. (30) Door de variabelen te wijzigen kun je dus de waarden optellen van de 
opgegeven rijnummers plus alle waarden die daartussen liggen.

• D5: =SOM(INDIRECT($G$4&$G$2):INDIRECT($G$4&$G$3))*INDIRECT("$G$5")  Doet 
hetzelfde als hierboven maar vermenigvuldigd de uitkomst met de variabele uit G5.

• D6: =INDIRECT("C"&G2)*SOM(B2:B7)  Vermenigvuldigt C2  met de som van de getallen 
B2:B7 (210).

Hieronder vind je nog een aantal voorbeelden van de functie INDIRECT(), genest met VERT.ZOEKEN() 
en als tweede voorbeeld  INDEX() >  VERGELIJKEN() met INDIRECT(). De voorwaarde voor een goed 
gebruik van de functie VERT.ZOEKEN() blijft wel staan, de kolom waarin de zoekwaarden zich 
bevinden moet oplopend gesorteerd zijn.

Zoeken in een gesorteerde tabel.

De formule in C8: =VERT.ZOEKEN(INDIRECT("A"&$D$2);A2:Z6;$E$2;WAAR)de zoekwaarde is 
variabel gemaakt. 

Door in E3 het getal 100 in te voeren, in E4 de letter A en in E5 de letter Z kun je de formule nog 
variabeler maken. De formule ziet er dan zo uit:
=VERT.ZOEKEN(INDIRECT("A"&$D$2);INDIRECT($E$4&$D$2):INDIRECT($E$5&$E$3);$E$2;WAAR)
Of het zinvol is om een formule zo variabel te maken laat ik aan jouw verbeelding over. Als de 
boodschap duidelijk is hoe je een formule variabel kunt maken is het mij goed. Missie geslaagd zullen 
we dan maar zeggen.

Een mooier en in de praktijk beter toepasbaar voorbeeld is het zoeken van een waarde in een ander 
werkblad. We gaan ervan uit dat je binnen een werkmap bladen hebt van verschillende jaren, die 
uiteraard hetzelfde van opbouw zijn. Nu wil je vanuit het huidige werkblad (2009) bezien wat het 
resultaat op dezelfde dag was in bijvoorbeeld 2008. Geen probleem want je kunt dat opvragen vanuit 
werkblad 2009. Voorwaarde voor VERT.ZOEKEN(): de kolom met de zoekwaarden is oplopend 
gesorteerd.

In het huidige werkblad (2009) geef je de criteria op, hier dus B9 en C9. In het werkblad waarin je wilt 
zoeken maak je een verwijzing naar C9 van het blad 2009  ='2009'!C9.
De formule in E6: =VERT.ZOEKEN(INDIRECT(B9&"!$B1");INDIRECT(B9&"!$A$2:$I$40");$E$2) De 
zoek en tabelmatrix zijn variabel gemaakt door INDIRECT(), de kolomindex is variabel vanwege de 
(absolute) celverwijzing.

NB, 
Het rijnummer uit het voorbeeld heeft in dit geval geen betekenis.
Het laatste argument van VERT.ZOEKEN gemist in de formule? Als benaderen WAAR is of wordt  
weggelaten, wordt er een exact of een niet-exact overeenkomende waarde gevonden. Wanneer er  

geen exacte overeenkomst wordt gevonden, wordt de volgende hoogste waarde die kleiner is dan  
zoekwaarde als resultaat gegeven. Hier is het argument dus weggelaten.

Zoeken in een ongesorteerde tabel.

Bij een ongesorteerd zoekbereik gebruiken we de formule: 
=INDEX(A2:C6;VERGELIJKEN(INDIRECT("A"&$D$2);A2:A6;0);$E$2)

De formule voor het zoeken in een ander werkblad:
=INDEX(INDIRECT(B9&"!$A$2:$C$6");VERGELIJKEN($C$9;INDIRECT($B$9&"!$A$2:A6");0);$E$2)

Zoeken in een andere werkmap.

Zoeken in een andere werkmap behoort ook tot de mogelijkheden. De formule kan er dan zo uitzien:
=VERT.ZOEKEN(INDIRECT(A10&'[Eneco 2008.xlsx]Overzicht'!$O$6). Het pad naar de betreffende 
werkmap (Eneco2008) en het werkblad (Overzicht) zijn in de formule aanwezig. 
De werkmap waarin je wilt zoeken moet geopend zijn, zo niet dan krijg je een foutmelding (#VERW!).

Pagina 3/3

Helpmij.nl handleiding: Excel: de functie INDIRECT()

 

Help
mij.n

l


